
	Reason for change:
	In the current TTCN implementation of the function f_TC_22_6_1_NBIOT, for the case when pc_NB_MultiDRB is set to TRUE, the current TTCN expects the PDN Connectivity REQ (Steps 16a4-16a5) on Ncell 50 and SRB1bis which is hardcoded. Similarly, at Step 16a11-16a13 the PDN Disconnect REQ is also expected on Ncell 50 and on SRB1bis, which is hardcoded. The Cell ID needs to be changed to v_CellID in both instances.

Also, as per 36.331 Sec 5.3.1.4, Once the Security is activated, the RRC Messages will be sent on SRB1. At Steps (16a4-16a5) and (16a11-16a13) the Security is already activated, and hence the messages should be expected on SRB1, not SRB1bis. This also needs to be corrected.

	
	

	Summary of change:
	Replace nbiot_Cell50 with v_CellID, and tsc_SRB1bis with tsc_SRB1 for Steps 16a4-16a5) and (16a11-16a13).

2. Corrections required for NBIOT test case 22.6.1
Change 1 f_TC_22_6_1_NBIOT ()
	Function name
	f_TC_22_6_1_NBIOT ()

	Reason for change
	In the current TTCN implementation of the function f_TC_22_6_1_NBIOT, for the case when pc_NB_MultiDRB is set to TRUE, the current TTCN expects the PDN Connectivity REQ (Steps 16a4-16a5) on Ncell 50 and SRB1bis which is hardcoded. Similarly, at Step 16a11-16a13 the PDN Disconnect REQ is also expected on Ncell 50 and on SRB1bis, which is hardcoded. The Cell ID needs to be changed to v_CellID in both instances.

Also, as per 36.331 Sec 5.3.1.4, Once the Security is activated, the RRC Messages will be sent on SRB1. At Steps (16a4-16a5) and (16a11-16a13) the Security is already activated, and hence the messages should be expected on SRB1, not SRB1bis. This also needs to be corrected.

	Summary of change
	Replace nbiot_Cell50 with v_CellID, and tsc_SRB1bis with tsc_SRB1 for Steps 16a4-16a5) and (16a11-16a13).

	TTCN module
	NBIOT_ESM_Testcases.ttcn

	MCC160 Comment
	


Before Change:

	<<SKIPPED CODE>>
if (pc_NB_MultiDRB) {
      f_UT_RequestAdditionalPDNConnectivity(UT, hex2int(v_EpsBearerId2), tsc_APN1_String);
      //@siclog "Step 16a4 - 16a5" siclog@
      SRB.receive(car_NB_SRB_NasPdu_IND(nbiot_Cell50,
                                        tsc_SRB1bis,
                                        cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                          cr_PDNConnReqAdditionalPDN(f_GetPdnType(), cr_AccessPointName(v_APN_Value)))))-> value v_ReceivedAsp;  /* @sic R5-176884 sic@ */
      v_PDNConnReq := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.pDN_CONNECTIVITY_REQUEST;
      // Get Procedure transaction identifier assigned by the UE
      v_PTId_UE := v_PDNConnReq.procedureTransactionIdentifier;
      v_AccessPointName := cs_AccessPointName(v_APN_Value);
      v_Pco := v_PDNConnReq.extdProtocolConfigurationOptions;
      //@siclog "Step 16a6 - 16a8" siclog@
      v_PDN_Address2 := f_EUTRA_NB_GetPDNAddress(f_CheckExtdPCOforIPallocationViaNas(v_Pco), PDN_2);
      f_NBIOT_ActivateDefaultEpsBearerContext_UP(v_CellId,
                                                 v_EpsBearerId2,
                                                 v_PTId_UE,  // use value as defined by the UE
                                                 v_AccessPointName,
                                                 v_PDN_Address2,
                                                 f_GetDefaultExtdProtocolConfigOptions(v_Pco));
      f_NBIOT_DelayForUserPlaneSignalling_UP(v_CellId, cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                                         cr_508_ActivateDefEpsBearerContextAccept(v_EpsBearerId2)));
      //@siclog "Step 16a9 Void" siclog@
      //@siclog "Step 16a10" siclog@
      f_UT_DeactivateBearer(UT, hex2int(v_EpsBearerId2));
      //@siclog "Step 16a11 - 16a13" siclog@
      SRB.receive(car_NB_SRB_NasPdu_IND(nbiot_Cell50,
                                        tsc_SRB1bis,
                                        cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                          cr_PDNDisconnectReq(cr_LinkedId(hex2bit(v_EpsBearerId2))))))-> value v_ReceivedAsp; // @sic R5-176884 sic@
      f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 16a12");
      // !!!! PROSE ISSUE: Don't need to configure any DRBs here !!!!
      //@siclog "Step 16a14 - 16a15" siclog@
<<SKIPPED CODE>>


After Change:

	<<SKIPPED CODE>>
if (pc_NB_MultiDRB) {       
      f_UT_RequestAdditionalPDNConnectivity(UT, hex2int(v_EpsBearerId2), tsc_APN1_String);      
      //@siclog "Step 16a4 - 16a5" siclog@
      SRB.receive(car_NB_SRB_NasPdu_IND(v_CellId, //WA#22_6_1 nbiot_Cell50,
                                        tsc_SRB1, //WA#22_6_1 tsc_SRB1bis,
                                        cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                          cr_PDNConnReqAdditionalPDN(f_GetPdnType(), cr_AccessPointName(v_APN_Value)))))-> value v_ReceivedAsp;  /* @sic R5-176884 sic@ */
      v_PDNConnReq := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.pDN_CONNECTIVITY_REQUEST;      
      // Get Procedure transaction identifier assigned by the UE
      v_PTId_UE := v_PDNConnReq.procedureTransactionIdentifier;
      v_AccessPointName := cs_AccessPointName(v_APN_Value);
      v_Pco := v_PDNConnReq.extdProtocolConfigurationOptions;
      //@siclog "Step 16a6 - 16a8" siclog@
      v_PDN_Address2 := f_EUTRA_NB_GetPDNAddress(f_CheckExtdPCOforIPallocationViaNas(v_Pco), PDN_2);
      f_NBIOT_ActivateDefaultEpsBearerContext_UP(v_CellId,
                                                 v_EpsBearerId2,
                                                 v_PTId_UE,  // use value as defined by the UE
                                                 v_AccessPointName,
                                                 v_PDN_Address2,
                                                 f_GetDefaultExtdProtocolConfigOptions(v_Pco));
      f_NBIOT_DelayForUserPlaneSignalling_UP(v_CellId, cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                                         cr_508_ActivateDefEpsBearerContextAccept(v_EpsBearerId2)));
      //@siclog "Step 16a9 Void" siclog@
      //@siclog "Step 16a10" siclog@
      f_UT_DeactivateBearer(UT, hex2int(v_EpsBearerId2));
      //@siclog "Step 16a11 - 16a13" siclog@
      SRB.receive(car_NB_SRB_NasPdu_IND(v_CellId, //WA#22_6_1 nbiot_Cell50,
                                        tsc_SRB1, //WA#22_6_1 tsc_SRB1bis,
                                        cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                          cr_PDNDisconnectReq(cr_LinkedId(hex2bit(v_EpsBearerId2))))))-> value v_ReceivedAsp; // @sic R5-176884 sic@
      f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 16a12");
      // !!!! PROSE ISSUE: Don't need to configure any DRBs here !!!!
      //@siclog "Step 16a14 - 16a15" siclog@
<<SKIPPED CODE>>


Change 2 f_NBIOT_DelayForUserPlaneSignalling_UP
	Function name
	f_NBIOT_DelayForUserPlaneSignalling_UP

	Reason for change
	In the current TTCN implementation of the function f_TC_22_6_1_NBIOT, for the case when pc_NB_MultiDRB is set to TRUE, the function f_NBIOT_DelayForUserPlaneSignalling_UP is called. In this function, current TTCN expects the UE to transmit the Activate Default EPS Bearer Context, and the subsequent L2 Data IND on DRB1.

However, for the testcase 22.6.1, the DRB1 is already associated with the first Default Bearer context. Hence, the UE transmits this L2 Data Ind on the DRB2, which is unhandled. 

This needs to be included in TTCN.

	Summary of change
	Added the handling for L2 Data IND on DRB2. Please see screenshot below.

	TTCN module
	NBIOT_ESM_Testcases.ttcn

	MCC160 Comment
	


Before Change:

	function f_NBIOT_DelayForUserPlaneSignalling_UP(NBIOT_CellId_Type p_CellId,
                                                  template NAS_MSG_Indication_Type p_NAS_MSG_Indication) runs on NBIOT_PTC
  {
    var NB_L2_DATA_IND v_NB_L2_DATA_IND;
    var octetstring v_DataReply;
    var boolean v_WaitingForUserPlaneSignalling := true;
    var boolean v_WaitingForNasSignalling := ispresent(p_NAS_MSG_Indication);
    timer t_WaitForIPSignalling;
    t_WaitForIPSignalling.start(1.2);
    alt {
      [v_WaitingForUserPlaneSignalling] t_WaitForIPSignalling.timeout {
        v_WaitingForUserPlaneSignalling := false;
        if (v_WaitingForNasSignalling) {
          repeat;
        }
      }
      [v_WaitingForUserPlaneSignalling] L2DATA.receive(car_NB_L2_DATA_IND(cr_NB_IndAspCommonPart_RB(p_CellId, tsc_NB_RbId_DRB1), ?)) -> value v_NB_L2_DATA_IND {
        v_DataReply := f_NBIOT_ProcessIPv6Address(v_NB_L2_DATA_IND.L2Data.SubframeData.PduSduList.PdcpSdu[0]);
        L2DATA.send(cas_NB_L2_DATA_REQ(cs_NB_ReqAspCommonPart_RB(p_CellId, tsc_NB_RbId_DRB1), cs_L2Data_Request_PdcpSdu(v_DataReply)));
        repeat; // @sic R5s170748 sic@
      }
      [v_WaitingForNasSignalling] SRB.receive(car_NB_SRB_NasPdu_IND(p_CellId, tsc_SRB1, p_NAS_MSG_Indication)) {  // compiler warning cannot be avoided
        v_WaitingForNasSignalling := false;
        if (v_WaitingForUserPlaneSignalling) {
          repeat;
        }
      }
    }
  }


After Change:

	function f_NBIOT_DelayForUserPlaneSignalling_UP(NBIOT_CellId_Type p_CellId,
                                                  template NAS_MSG_Indication_Type p_NAS_MSG_Indication) runs on NBIOT_PTC
  {
    var NB_L2_DATA_IND v_NB_L2_DATA_IND;
    var octetstring v_DataReply;
    var boolean v_WaitingForUserPlaneSignalling := true;
    var boolean v_WaitingForNasSignalling := ispresent(p_NAS_MSG_Indication);
    timer t_WaitForIPSignalling;
    t_WaitForIPSignalling.start(1.2);
    alt {
      [v_WaitingForUserPlaneSignalling] t_WaitForIPSignalling.timeout {
        v_WaitingForUserPlaneSignalling := false;
        if (v_WaitingForNasSignalling) {
          repeat;
        }
      }
      [v_WaitingForUserPlaneSignalling] L2DATA.receive(car_NB_L2_DATA_IND(cr_NB_IndAspCommonPart_RB(p_CellId, tsc_NB_RbId_DRB1), ?)) -> value v_NB_L2_DATA_IND {
        v_DataReply := f_NBIOT_ProcessIPv6Address(v_NB_L2_DATA_IND.L2Data.SubframeData.PduSduList.PdcpSdu[0]);
        L2DATA.send(cas_NB_L2_DATA_REQ(cs_NB_ReqAspCommonPart_RB(p_CellId, tsc_NB_RbId_DRB1), cs_L2Data_Request_PdcpSdu(v_DataReply)));
        repeat; // @sic R5s170748 sic@
      }
      //WA#22_6_1 : Start
      [v_WaitingForUserPlaneSignalling] L2DATA.receive(car_NB_L2_DATA_IND(cr_NB_IndAspCommonPart_RB(p_CellId, tsc_NB_RbId_DRB2), ?)) -> value v_NB_L2_DATA_IND {
        v_DataReply := f_NBIOT_ProcessIPv6Address(v_NB_L2_DATA_IND.L2Data.SubframeData.PduSduList.PdcpSdu[0]);
        L2DATA.send(cas_NB_L2_DATA_REQ(cs_NB_ReqAspCommonPart_RB(p_CellId, tsc_NB_RbId_DRB2), cs_L2Data_Request_PdcpSdu(v_DataReply)));
        repeat; // @sic R5s170748 sic@
      }
      //WA#22_6_1 : Stop
      [v_WaitingForNasSignalling] SRB.receive(car_NB_SRB_NasPdu_IND(p_CellId, tsc_SRB1, p_NAS_MSG_Indication)) {  // compiler warning cannot be avoided
        v_WaitingForNasSignalling := false;
        if (v_WaitingForUserPlaneSignalling) {
          repeat;
        }
      }
    }
  }


